
This paper describes the FunnyQT solution to the TTC 2015 Model Execution transformation case.
The solution solves the third variant of the case, i.e., it considers and implements the execution
semantics of the complete UML Activity Diagram language.

FunnyQT is a model querying and model transformation library for the functional Lisp-dialect
Clojure providing a comprehensive and efficient querying and transformation API, many parts of
which are provided as task-oriented embedded DSLs.

1 Introduction

This paper describes the FunnyQT1 [2, 3] solution of the TTC 2015 Model Execution Case [4]. It
implements the third variant of the case description, i.e., it implements the execution semantics of the
complete UML Activity Diagram language. The solution project is available on Github2, and it is set up
for easy reproduction on a SHARE image3.

FunnyQT is a model querying and transformation library for the functional Lisp dialect Clojure4.
Queries and transformations are plain Clojure programs using the features provided by the FunnyQT
API.

As a Lisp, Clojure provides strong metaprogramming capabilities that are exploited by FunnyQT
in order to define several embedded domain-specific languages (DSL, [1]) for different querying and
transformation tasks.

FunnyQT is designed with extensibility in mind. By default, it supports EMF [5] models and
JGraLab5 TGraph models. Support for other modeling frameworks can be added without having to
touch FunnyQT’s internals.

The FunnyQT API is structured into the following namespaces, each namespace providing constructs
supporting concrete querying and transformation use-cases:
funnyqt.emf EMF-specific model management API
funnyqt.tg JGraLab/TGraph-specific model management API
funnyqt.generic Protocol-based, generic model management API
funnyqt.query Generic querying constructs such as quantified expressions or regular path expressions
funnyqt.polyfns Constructs for defining polymorphic functions dispatching on metamodel types
funnyqt.pmatch Pattern matching constructs
funnyqt.relational Constructs for logic-based, relational model querying inspired by Prolog
funnyqt.in-place In-place transformation rule definition constructs

1http://funnyqt.org
2https://github.com/tsdh/ttc15-model-execution-funnyqt
3The SHARE image name is ArchLinux64_TTC15-FunnyQT_2
4http://clojure.org
5http://jgralab.github.io

funnyqt.model2model Out-place transformation definition constructs similar to ATL or QVT Opera-
tional Mappings

funnyqt.extensional Transformation API similar to GReTL
funnyqt.bidi Constructs for defining bidirectional transformations similar to QVT Relations
funnyqt.coevo Constructs for transformations that evolve a metamodel and a conforming model simul-

taneously at runtime
funnyqt.visualization Model visualization
funnyqt.xmltg Constructs for querying and modifying XML files as models conforming to a DOM-like

metamodel
For solving the model execution case, the funnyqt.emf, funnyqt.query, and funnyqt.polyfns names-

paces have been used.

2 Solution Description

The explanations in the case description about the operational semantics on UML Activity Diagrams
suggest an algorithmic solution to the transformation case. The FunnyQT solution tries to be almost a
literal translation of the case description to Clojure code.

The first line of the solution calls the generate-ecore-model-functions FunnyQT macro.
1 (generate-ecore-model-functions "activitydiagram.ecore" ttc15-model-execution-funnyqt.ad a)

As its name suggests, it generates a metamodel-specific API. This API is generated into the names-
pace ttc15-model-execution-funnyqt.ad, and the namespace alias a is used to refer to that namespace
from the current one.

The generated API consists of element creation functions, lazy element sequence functions, attribute
access functions, and reference access functions. For example, (a/create-ControlToken! ad) creates a
new control token and adds it to the activity diagram model ad, (a/isa-Token? x) returns true if and only
if x is a token, (a/all-Inputs ad) returns the lazy sequence of input elements in ad, (a/running? n)
and (a/set-running! n true) query and set the node n’s running attribute, and (a/->locals a),
(a/->set-locals! a ls), (a/->add-locals! a l), and (a/->remove-locals! a l) query, set, add
to, and remove from the locals reference of the activity a.

Metamodel-specific APIs created by the generate-ecore-model-functions macro are independent
of the underlying implementation kind. The FunnyQT solution uses a dynamic instance model but it
would work equivalently if it was run with an EMF model using classes generated for the metamodel’s
generator model.

Instead of using the generated API, the solution could also use the generic EMF API in which case the
lazy sequence of a activity diagram models inputs would be retrieved using (eallcontents ad ’Input).
But the generated API is slightly more concise and readable.

In the following, the solution is presented in a top-down manner similar to how the case description
defines the operational semantics of activity diagrams. Listing 1 on the facing page shows the function
execute-activity-diagram which contains the transformation’s main loop.

The function queries the single activity in the diagram, creates a new trace, and assigns that to the
activity. Then, the activity’s variables are initialized and its nodes are set running.

Then, a loop-recur iteration6 performs the actual execution of the activity. The variable en is bound
to the first enabled node. Initially, this is the activity’s initial node. As long as there is an enabled node

6loop is not a loop in the sense of Java’s for or while but a local tail-recursion construct. The loop declares variables

2 (defn execute-activity-diagram [ad]
3 (let [activity (the (a/all-Activities ad))
4 trace (a/create-Trace! nil)]
5 (a/->set-trace! activity trace)
6 (init-variables activity (first (a/all-Inputs ad)))
7 (mapc #(a/set-running! % true) (a/->nodes activity))
8 (loop [en (first (filter a/isa-InitialNode? (a/->nodes activity)))]
9 (when en

10 (exec-node en)
11 (a/->add-executedNodes! trace en)
12 (recur (first (enabled-nodes activity)))))
13 trace))

Listing 1: The main loop of executing activity diagrams

left, it gets executed and then added to the trace. Thereafter, the loop is restarted with the next enabled
node. Eventually, there won’t be an enabled node left, and then the function returns the trace.

So the first step in the execution of an activity is the initialization of its local and input variables.
The corresponding function init-variables is shown in listing 2. For locals, their current value is set
to their initial value if there is one defined. For input variables, their current value is set to the value of
the input’s corresponding input value element.

14 (defn init-variables [activity input]
15 (doseq [lv (a/->locals activity)]
16 (when-let [init-value (a/->initialValue lv)]
17 (a/->set-currentValue! lv init-value)))
18 (doseq [iv (and input (a/->inputValues input))]
19 (when-let [val (a/->value iv)]
20 (a/->set-currentValue! (a/->variable iv) val))))

Listing 2: Initialization of variables

After initializing the variables, the main function sets the activity’s nodes running, and the main loop
starts with the activity’s initial node which will be executed.

For different kinds of activity nodes, different execution semantics have to be encoded. This is exactly
the use-case of FunnyQT’s polymorphic functions (polyfn). A polymorphic function is declared once,
and then different implementations for instances of different metamodel types can be defined. When
the polyfn is called, a polymorphic dispatch based on the polyfn’s first argument’s metamodel type is
performed to pick out the right implementation7.

Listing 3 on the following page shows the declaration of the polyfn exec-node and its implementation
for initial nodes. The declaration only defines the name of the polyfn and the number of its arguments
(just one, here). The implementation for initial nodes simply offers one new control token to the initial
node’s outgoing control flow edge.8

Listing 4 on the next page shows the exec-node implementations for join, merge, and decision nodes.
Join and Merge nodes simply consume their input offers and pass the tokens they have been offered on
all outgoing control flows. Decision nodes act similar but offer their input tokens only on the outgoing

with their initial bindings, and in the loop’s body recur forms may recurse back to the beginning of the loop providing new
bindings for the loop’s variables.

7Polyfns support multiple inheritance. In case of an ambiguity during dispatch, e.g., two or more inherited implementations
are applicable, an error is signaled.

8The FunnyQT function the is similar to Clojure’s first except that it signals an error if the given collection contains
zero or more than one element. Thus, it makes the assumption that there must be only one outgoing control flow explicit.

21 (declare-polyfn exec-node [node])

22 (defn offer-one-ctrl-token [node]
23 (let [ctrl-t (a/create-ControlToken! nil)
24 out-cf (the (a/->outgoing node))
25 offer (a/create-Offer! nil {:offeredTokens [ctrl-t]})]
26 (a/->add-heldTokens! node ctrl-t)
27 (a/->add-offers! out-cf offer)))

28 (defpolyfn exec-node InitialNode [i]
29 (offer-one-ctrl-token i))

Listing 3: Declaration of polyfn exec-node and its implementation for initial nodes

control flow whose guard variable’s current value is true9.

30 (defn pass-tokens
31 ([n] (pass-tokens n nil))
32 ([n out-cf]
33 (let [in-toks (consume-offers n)]
34 (a/->set-heldTokens! n in-toks)
35 (doseq [out-cf (if out-cf [out-cf] (a/->outgoing n))]
36 (a/->add-offers!
37 out-cf (a/create-Offer!
38 nil {:offeredTokens in-toks}))))))

39 (defpolyfn exec-node JoinNode [jn]
40 (pass-tokens jn))

41 (defpolyfn exec-node MergeNode [mn]
42 (pass-tokens mn))

43 (defpolyfn exec-node DecisionNode [dn]
44 (pass-tokens dn (the #(-> % a/->guard a/->currentValue a/value)
45 (a/->outgoing dn))))

Listing 4: exec-node impls for join, merge, and decision nodes

So how does a node consume offers? This is defined by the consume-offers function shown in
listing 5 on the facing page. First, the offers and the tokens offered by them are calculated. Then, the
offered tokens are divided into control and forked tokens. For control tokens, their holder is unset. For
forked tokens, the corresponding base token’s holder is unset. The forked tokens’ remainingOffersCount
is decremented. If it has become zero then, the forked token is removed from its holder. Lastly, the offers
are deleted, and the incoming tokens are returned.

The remaining kinds of activity nodes are fork nodes, activity final nodes and opaque actions. Their
exec-node implementations are printed in listing 6 on the next page.

A fork node consumes its offers and creates one forked token per incoming token. The incoming
tokens are set as the forked tokens’ base tokens, and the remaining offers count is set to the number of
outgoing control flows. All created forked tokens are offered on each outgoing control flow.

An activity final node simply consumes all offers and then sets the running attribute of all nodes
contained by the executed activity to false. An opaque action also consumes all offers, then evaluates all
its expressions in sequence using the eval-exp function, and finally offers one single control token on
the outgoing control flow.

9(the predicate collection) returns the single element of the collection for which the predicate returns true. If
there is no or more elements satisfying the predicate, an error is signaled.

46 (defn consume-offers [node]
47 (let [offers (mapcat a/->offers (a/->incoming node))
48 tokens (mapcat a/->offeredTokens offers)
49 ctrl-toks (filter a/isa-ControlToken? tokens)
50 fork-toks (filter a/isa-ForkedToken? tokens)]
51 (doseq [ct ctrl-toks]
52 (a/->set-holder! ct nil))
53 (doseq [ft fork-toks]
54 (when-let [bt (a/->baseToken ft)]
55 (a/->set-holder! bt nil))
56 (a/set-remainingOffersCount! ft (dec (a/remainingOffersCount ft)))
57 (when (zero? (a/remainingOffersCount ft))
58 (a/->set-holder! ft nil)))
59 (mapc edelete! offers)
60 tokens))

Listing 5: Consuming offers

61 (defpolyfn exec-node ForkNode [fn]
62 (let [in-toks (consume-offers fn)
63 out-cfs (a/->outgoing fn)
64 out-toks (mapv #(a/create-ForkedToken!
65 nil {:baseToken %, :holder fn,
66 :remainingOffersCount (count out-cfs)})
67 in-toks)]
68 (a/->set-heldTokens! fn in-toks)
69 (doseq [out-cf out-cfs]
70 (a/->add-offers! out-cf (a/create-Offer!
71 nil {:offeredTokens out-toks})))))
72 (defpolyfn exec-node ActivityFinalNode [afn]
73 (consume-offers afn)
74 (mapc #(a/set-running! % false)
75 (-> afn a/->activity a/->nodes)))

76 (defpolyfn exec-node OpaqueAction [oa]
77 (consume-offers oa)
78 (mapc eval-exp (a/->expressions oa))
79 (offer-one-ctrl-token oa))

Listing 6: exec-node impls for activity final nodes and opaque actions

How an expression is evaluated depends on (1) the type of the and (2) on the value of its operator
attribute. The expression’s type is only important in order to separate unary from binary expressions,
and the operator defines the semantics. Therefore, the eval-exp function shown in listing 7 on the
following page has a special case for boolean unary expressions which negates the expression’s current
value using not. For all binary expressions, the map op2fn mapping from operator enum constants to
Clojure functions having the semantics of that operator is used. The function determined by looking up
the expression’s operator is applied to both operands to compute the new value.

After executing all enabled nodes, the transformation’s main function execute-activity-diagram
from listing 1 on page 3 recomputes the enabled nodes and resumes the execution. The enabled nodes are
computed by the enabled-nodes function shown in listing 8 on the following page. The enabled nodes
are those nodes of a given activity which are set running, are no initial nodes10, and receive an offer on
each incoming control flow, or, in the case of a merge node, on one incoming control flow.

These 102 NCLOC of algorithmic FunnyQT/Clojure code implement the complete operational se-
mantics of UML Activity Diagrams (with the exception of data flows which has not been demanded by
the case description).

10Initial nodes have to be excluded because if they are set running, all of their (zero) incoming control flows have offers.

80 (def op2fn {(a/enum-IntegerCalculationOperator-ADD) +
81 (a/enum-IntegerCalculationOperator-SUBRACT) -
82 (a/enum-IntegerComparisonOperator-SMALLER) <
83 (a/enum-IntegerComparisonOperator-SMALLER_EQUALS) <=
84 (a/enum-IntegerComparisonOperator-EQUALS) =
85 (a/enum-IntegerComparisonOperator-GREATER_EQUALS) >=
86 (a/enum-IntegerComparisonOperator-GREATER) >
87 (a/enum-BooleanBinaryOperator-AND) #(and %1 %2)
88 (a/enum-BooleanBinaryOperator-OR) #(or %1 %2)})

89 (defn eval-exp [exp]
90 (a/set-value! (-> exp a/->assignee a/->currentValue)
91 (if (a/isa-BooleanUnaryExpression? exp)
92 (not (-> exp a/->operand a/->currentValue a/value))
93 ((op2fn (a/operator exp))
94 (-> exp a/->operand1 a/->currentValue a/value)
95 (-> exp a/->operand2 a/->currentValue a/value)))))

Listing 7: Evaluation of expressions

96 (defn enabled-nodes [activity]
97 (filter (fn [n]
98 (and (a/running? n)
99 (not (a/isa-InitialNode? n))

100 ((if (a/isa-MergeNode? n) exists? forall?)
101 #(seq (a/->offers %)) (a/->incoming n))))
102 (a/->nodes activity)))

Listing 8: Computation of enabled nodes

3 Evaluation

In this section, the FunnyQT solution is evaluated according to the criteria stated in the case description.

Correctness. The JUnit test suite of the reference Java solution has been translated to Clojure and its
unit testing library. The very same assertions are tested, i.e., the execution order of nodes is tested and the
final values of variables are checked. Additionally, it is checked that after an activity has been executed,
there are no tokens leftover whose existence would hint at some bug in the implementation.

All tests pass for all provided test models, so the solution is correct at least with respect to these
models and the set of assertions tested by the unit test.

Understandability and conciseness. With 102 lines of non-commented source code, the FunnyQT
solution is quite concise.

Of course, understandability is a very subjective measure measure. The solution should be evident
for any Clojure programmer but even without prior Clojure knowledge, the solution shouldn’t be hard
to follow due to the usage of the metamodel-specific API. Another strong point is that all steps in the
execution of an activity are encoded in one function each whose definition is almost a literal translation
of the English description to FunnyQT/Clojure code.

Performance. Table 1 on the next page shows the execution times of the FunnyQT solution for all
provided test models. These times were measured on a normal 4-core laptop with 2.6 GHz and 2 GB of
RAM dedicated to the JVM.

Model Execution time
test1 1.3 ms
test2 0.6 ms
test3 4.1 ms
test4 3.2 ms
test5 0.5 ms
test6 (false) 3.7 ms
test6 (true) 5.4 ms
test-performance-variant-1 1104.0 ms
test-performance-variant-2 1246.5 ms
test-performance-variant-3-1 1159.6 ms
test-performance-variant-3-2 72.7 ms

Table 1: Execution times for the provided test models

When compared with the reference Java solution, the FunnyQT solution is slightly faster for all
normal and performance test models. However, for the model test-performance-variant-3-1, the Java
solution takes about 8 seconds. So for this model, the FunnyQT solution is about seven times faster than
the Java solution which hints at some inefficiencies in how parallel branches are executed in the the Java
solution.

4 Conclusion

In this paper, the FunnyQT solution to the TTC 2015 Model Execution case has been discussed. The
solution implements the full operational semantics of UML Activity Diagrams with the exception of
object flows which haven’t been considered in this case.

The specification is very concise. The complete implementation amounts to only 102 lines of code
without counting comments, empty lines, and the solution namespace’s namespace declaration (similar
to Java’s package and import statements).

The specification is also well-understandable. The description of the semantics given in the case
description have been translated almost literally to functions and one polymorphic function. These func-
tions use a metamodel-specific API which FunnyQT has generated from the activity diagram metamodel.

The FunnyQT solution also performs very well. In general, its execution times are very similar to
those of the Java reference solution, and for two of the performance test models, the FunnyQT solution
is in fact several times faster.

Overall, FunnyQT seems to be very adequate for defining model interpreters. Especially its poly-
morphic function facility has been explicitly designed for these kinds of tasks.

References

[1] Martin Fowler (2010): Domain-Specific Languages. Addison-Wesley Professional.

[2] Tassilo Horn (2013): Model Querying with FunnyQT - (Extended Abstract). In Keith Duddy & Gerti Kappel,
editors: ICMT, Lecture Notes in Computer Science 7909, Springer, pp. 56–57.

[3] Tassilo Horn (2015): Graph Pattern Matching as an Embedded Clojure DSL. In: International Conference on
Graph Transformation - 8th International Conference, ICGT 2015, L’Aquila, Italy, July 2015. To appear.



8 Solving the TTC Model Execution Case with FunnyQT

[4] Tanja Mayerhofer & Manuel Wimmer (2015): The TTC 2015 Model Execution Case. In: Transformation Tool
Contest 2015.

[5] Dave Steinberg, Frank Budinsky, Marcelo Paternostro & Ed Merks (2008): EMF: Eclipse Modeling Frame-
work, 2 edition. Addison-Wesley Professional.


	Introduction
	Solution Description
	Evaluation
	Conclusion

